Les valeurs extrêmes: Méthode des multiplicateurs de Lagrange

(Partie 3 de 3)

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en mars 2006. Cette feuille Maple a pour but de présenter la **méthode des multiplicateurs de Lagrange**. Après une justification de la méthode à l'aide d'un traitement graphique d'un premier exemple d'optimisation, deux autres exemples sont développés de façon algébrique pour optimiser une fonction de deux variables avec une contrainte d'égalité. La **méthode des multiplicateurs de Lagrange** est particulièrement utile lorsque la fonction ne peut être ramenée à une fonction de deux variables sans contrainte. Rappelons la condition nécessaire d'applicabilité de cette méthode: le gradient de la fonction contrainte ne doit aucunement s'annuler en quelque valeurs (x, y).

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,contourplot,contourplot3d,display,implicitplot,
  pointplot3d,setoptions,setoptions3d,spacecurve,textplot):
  setoptions(size=[300,300],labels=[x,y],axesfont=[TIMES,ROMAN,8],
  labelfont=[TIMES,ROMAN,8]):
  setoptions3d(size=[400,400],labels=[x,y,z],axes=frame,axesfont=
  [TIMES,ROMAN,8],labelfont=[TIMES,ROMAN,8]):
> with(plottools,arrow,transform):
> with(VectorCalculus,Del):
> macro(navy_pâle=COLOR(RGB, .0, .49019, .99215)):
```

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimamt les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de
  composition étendue
  Typesetting:-Settings(striptrailing=true);
```

$$extended$$

$$false \quad (1.1)$$

```
> _EnvExplicit:=true:
```

Présentation de la méthode

Considérons la fonction f définie par $f(x, y) = x^3 + xy + 2y^3$.

```
> f:=(x,y)->x^3+x*y+2*y^3:
```

```
'f'(x,y)=f(x,y);
```

$$f(x,y) = x^3 + 2y^3 + yx \qquad \textbf{(2.1)}$$

Le domaine de la fonction f est $\mathbb{R}^2$. Proposons-nous de déterminer les valeurs extêmes que prend la fonction f lorsque les valeurs (x, y) de son domaine doivent satisfaire l'égalité $\frac{(x-9)^2}{4} + \frac{(y-7)^2}{9} = 1$.

Autrement dit, il s'agit de déterminer les valeurs extrêmes de f lorsque les valeurs (x, y) de son domaine parcourent une ellipse. Illustrons dans un même graphique la surface S d'équation $z = \mathrm{f}(x, y)$, l'ellipse dans le plan de corrdonnées Oxy ainsi que l'image de cette ellipse sur la surface S.

Afin de bien illustrer l'ellipse et son image, nous aurons recours à la paramétrisation de cette ellipse. Pour une ellipse d'équation

$$\frac{(x-h)^2}{a^2} + \frac{(y-k)^2}{b^2} = 1$$

des équations de paramétrisation sont

$$x = a\cos(t) + h$$
$$y = b\sin(t) + k$$

```
> Surface:=plot3d([x,y,f(x,y)],x=-5..12,y=-5..12):
Ellipse:=spacecurve([2*cos(t)+9,3*sin(t)+7,0],t=0..2*Pi,color=orange,
thickness=2):
Ellipse_S:=spacecurve([2*cos(t)+9,3*sin(t)+7,f(2*cos(t)+9,3*sin(t)+7)
],t=0..2*Pi,
                      color=navy,thickness=2):
n:=24:
L:=seq(plots[spacecurve]([[2*cos(k*Pi/n)+9,3*sin(k*Pi/n)+7,0],[2*cos
(k*Pi/n)+9,3*sin(k*Pi/n)+7,
                          f(2*cos(k*Pi/n)+9,3*sin(k*Pi/n)+7)]],color=
navy_pâle,thickness=1),k=0..2*n):
> display(Surface,Ellipse,Ellipse_S,L,lightmodel=none,style=
patchnogrid,orientation=[-40,80],axes=framed);
```



En modifiant manuellement l'orientation du graphique précédent, il est facile de voir que la fonction f atteint effectivement un maximum absolu et un minimum absolu lorsque les valeurs du domaine de f parcours l'ellipse.

Affichons de nouveau le graphique précédent mais avec le style contour.

```
> Surface:=plot3d([x,y,f(x,y)],x=-5..12,y=-5..12,style=contour,
  contours=18,color=orange):
  display(Surface,Ellipse,Ellipse_S,L,orientation=[-80,60],axes=framed)
  ;
```



Illustrons maintenant les courbes de niveau comme nous le faisons habituellement.

```
> Contours_3d:=contourplot3d(f(x,y),x=-5..12,y=-5..12,contours=18,
  color=orange):
  Contours_2d:=contourplot(f(x,y),x=-5..12,y=-5..12,filled=true,
  contours=18,coloring=[white,orange]):
  t:=transform((x,y)->[x,y,0]):
> display(Contours_3d,Ellipse,Ellipse_S,L,t(Contours_2d),axes=framed,
  orientation=[-80,60]);
```



Illustrons finalement les courbes de niveau de la fonction f en 2d en lui superposant le tracé de l'ellipse.

```
> Ellipse_2d:=plot([2*cos(t)+9,3*sin(t)+7,t=0..2*Pi],color=orange,
  thickness=2):
> display(Contours_2d,Ellipse_2d);
```



Dans le contexte d'un problème d'optimisation, la fonction f est appelée fonction objective tandis que l'ellipse est appelé la contrainte. Le problème d'optimisation consiste à localiser sur le chemin de l'ellipse où la fonction est à son maximum et à son minimum. Dans un cas comme dans l'autre, le graphique précédent suggère que le maximum et le minimum se produisent lorsque qu'une **courbe de niveau de la fonction objective est tangente à la courbe contrainte**. Voilà donc l'idée maîtresse de la méthode des multiplicateurs de Lagrange.

L'équation de l'ellipse permet de définir une fonction g telle que $g(x, y) = \frac{(x-9)^2}{4} + \frac{(y-7)^2}{9} - 1$. Ainsi, la contrainte spécifiant que les valeurs du domaine de la fonction doivent parcourir l'ellipse est traduite par l'égalité

$$g(x, y) = 0$$

```
> g:=(x,y)->(x-9)^2/4+(y-7)^2/9-1;
```

$$g := (x, y) \mapsto \frac{(x-9)^2}{4} + \frac{(y-7)^2}{9} - 1 \quad \textbf{(2.2)}$$

À l'endroit de la surface où la fonction atteint un minimum ou un maximum, la courbe de niveau à la fonction objective et la courbe contrainte sont tangentes, c'est-à-dire que ces deux courbes possèdent une tangente commune. Or, en tout point d'une courbe de niveau, le gradient est perpendiculaire à la courbe de niveau et, similairement, le gradient à la courbe contrainte est lui-même perpendiculaire à la courbe contrainte. Nous devons donc avoir pour (x, y), $grad\,f(x, y)$ grad f(x, y) et $grad\,g(x, y)$ parallèles donc, multiple scalaire.

$$grad\,f(x, y) = \lambda\ grad\,g(x, y)$$

Le paramètre « λ » est appelé multiplicateur de Lagrange. La méthode des multiplicateurs de Lagrange est une procédure algébrique permettant de déterminer les points maximum et minimum.

Pour déterminer (x, y), nous devons donc résoudre le système

$$grad\,f(x, y) = \lambda\ grad\,g(x, y)$$
$$g(x, y) = 0$$

L'égalité

$$grad\,f(x, y) = \lambda\ grad\,g(x, y)$$

est équivalente à l'égalité

$$grad\,f(x, y) - \lambda\ grad\,g(x, y) = 0$$

Formons alors la fonction F définie par $F(x, y) = f(x, y) + \lambda\,g(x, y)$.

```
> F:=(x,y)->f(x,y)+lambda*g(x,y);
```

$$F := (x, y) \mapsto f(x, y) + \lambda \cdot g(x, y) \quad \textbf{(2.3)}$$

```
> Gradient:=convert(Del(F(x,y),[x,y]),vector[row]);
```

$$Gradient := \left[\begin{array}{cc} 3x^2 + y + \lambda\left(\frac{x}{2} - \frac{9}{2}\right) & 6y^2 + x + \lambda\left(\frac{2y}{9} - \frac{14}{9}\right)\end{array}\right] \qquad \textbf{(2.4)}$$

Résolvons maintenant le système S: $\{F_x(x,y) = 0, F_y(x,y) = 0, g(x,y) = 0\}$.

```
> Éq1:=Gradient[1]=0;
  Éq2:=Gradient[2]=0;
  Éq3:=g(x,y)=0;
```

$$\acute{E}q1 := 3x^2 + y + \lambda\left(\frac{x}{2} - \frac{9}{2}\right) = 0$$

$$\acute{E}q2 := 6y^2 + x + \lambda\left(\frac{2y}{9} - \frac{14}{9}\right) = 0$$

$$\acute{E}q3 := \frac{(x-9)^2}{4} + \frac{(y-7)^2}{9} - 1 = 0 \qquad \textbf{(2.5)}$$

```
> Sol:={solve({Éq1,Éq2,Éq3},{x,y,lambda})}: # Les accolades extérieures
  évitent la redondance des solutions.
> map(evalf[10],Sol);
```

$$\{\{\lambda = -933.7022055, x = 9.615209677, y = 9.854542587\}, \{\lambda = 280.0114357, x = 7.6968484, y = 4.724249401\}, \{\lambda = -233.3543999 - 21.0393657\,I, x = 13.57734079 - 1.545310704\,I, y = 4.458895122 - 6.263094835\,I\}, \{\lambda = -233.3543999 + 21.0393657\,I, x = 13.57734079 + 1.545310704\,I, y = 4.458895122 + 6.263094835\,I\}, \{\lambda = -40.4686709 - 42.4853761\,I, x = 5.878039571 - 4.704304818\,I, y = -0.5013112501 + 4.405225865\,I\}, \{\lambda = -40.4686709 + 42.4853761\,I, x = 5.878039571 + 4.704304818\,I, y = -0.5013112501 - 4.405225865\,I\}\} \qquad \textbf{(2.6)}$$

La résolution de ce système nous révèle deux points critiques que nous appellerons P et Q.

```
> P:=[9.615209673,9.8545418]:
  Q:=[7.696848400,4.7242485]:
```

Esquissons dans un même graphique le tracé de la constrainte (l'ellipse) et deux courbes de niveau judicieusement choisies soit celles passant par les points P et Q. Ces courbes seront évidemment tangentes à l'ellipse puisque les points P et Q sont aussi des points de cet ellipse.

```
> Contours_2d:=contourplot(f(x,y),x=0..15,y=0..15,contours=[f(P[1],P[2]
  ),f(Q[1],Q[2])],color=orange):
  Point_P:=textplot([P[1]+0.5,P[2],"P"],align={ABOVE,RIGHT}):
  Point_Q:=textplot([Q[1]-0.5,Q[2]+.5,"Q"],align={BELOW,LEFT}):
  display(Contours_2d,Ellipse_2d,Point_P,Point_Q,scaling=constrained);
```



Montrons qu'effectivement les deux vecteurs gradients sont perpendiculaires aux deux tracés en chacun des points P et Q.

```
> Dir_gradient1:=eval(convert(Del(f(x,y),[x,y]),Vector[row]),[x=P[1],y=
  P[2]]);
  Dir_gradient2:=eval(convert(Del(f(x,y),[x,y]),Vector[row]),[x=Q[1],y=
  Q[2]]);
```

$$Dir\_gradient1 := \begin{bmatrix} 287.211313 & 592.2871742 \end{bmatrix}$$

$$Dir\_gradient2 := \begin{bmatrix} 182.4486744 & 141.6079917 \end{bmatrix} \quad \textbf{(2.7)}$$

```
> V_Grad1:=arrow([P[1],P[2]], [P[1]+(287.2113132/(600)),P[2]+
  (592.2871979/(600))], .2, 0.5, .2, color=orange):
  V_Grad2:=arrow([Q[1],Q[2]], [Q[1]+(182.4486749/(-200)),Q[2]+
  (141.6080201/(-200))], .2, 0.5, .2, color=orange):
```

```
> display(Contours_2d,Ellipse_2d,Point_P,Point_Q,V_Grad1,V_Grad2,
  scaling=constrained);
```



Les valeurs maximum et minimum de $f(x,y) = x^3 + xy + 2y^3$ sur l'ellipse d'équation $\frac{(x-9)^2}{4} + \frac{(y-7)^2}{9} = 1$ sont

```
> Maximum=f(P[1],P[2]);
  Minimum=f(Q[1],Q[2]);
```

$$Maximum = 2897.689732$$

$$Minimum = 703.2109848 \quad \textbf{(2.8)}$$

Deux autres exemples

Initialisons d'abord la procédure ci-dessous. Cette procédure automatise l'énumération des points à contrôler dans le cas d'un nombre pas trop grand de points. Vous pourriez tout aussi bien les énumérer manuellement et ne pas initialiser cette procédure.

Procédure Listage

```
> Listage:=proc(Ensemble::set)
  global x,y,lambda;
  local Sol,Points,k;
  Sol:=Ensemble;
  assign(Sol[1]);
  Points:=[x,y];
  unassign('x,y,lambda');
  for k from 2 to nops(Sol) do
  assign(Sol[k]);
  Points:=Points,[x,y];
  unassign('x,y,lambda');
  od;
  Points;
  end:
```

Exemple 1

Soit la fonction f définie par $\mathrm{f}(x,y) = 2x^2 - y + y^2 + 5$. Optimisons cette fonction sur un disque de rayon 1 centré à l'origine. Puisque la fonction f est continue sur un domaine fermé et borné, la fonction f atteint son maximum et son minimum avec certains points de son domaine.

Commençons par créer la fonction f.

```
> f:=(x,y)->2*x^2-y+y^2+5;
```

$$f := (x,y) \mapsto 2 \cdot x^2 - y + y^2 + 5 \quad \textbf{(3.2.1)}$$

Dans un même graphique, superposons le tracé de la surface d'équation $z = \mathrm{f}(x,y)$ et celui du domaine fermé d'optimisation.

Pour obtenir le tracé de la frontière du domaine, soit celui d'un cercle d'équation $x^2 + y^2 = 1$, paramétrisons cette frontière en posant $x = \cos(t)$ et $y = \sin(t)$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-2..2,y=-2..2):
  Contrainte:=spacecurve([sin(t),cos(t),0],t=0..2*Pi,color=navy,
  thickness=3):
  Domaine:=plot3d([x,y,0],x=-2..2,y=-2..2,style=patchnogrid,color=
  turquoise):
  Domaine_optimisation:=plot3d([x,y,0],x=-1..1,y=-sqrt(1-x^2)..sqrt
  (1-x^2),style=patchnogrid,color=navy):
  display([Surface,Contrainte,Domaine,Domaine_optimisation],axes=
  framed,orientation=[35,65]);
```



Pour mieux se convaincre que la fonction possède effectivement des extréma absolus avec des points du disque, traçons l'image de ce disque par la fonction f. Superposons au graphique quelques segments verticaux afin d'accentuer l'image de quelques valeurs de la frontière.

```
> Contrainte_S:=spacecurve([sin(t),cos(t),f(sin(t),cos(t))],t=0..2*
  Pi,color=navy,thickness=2):
  n:=48:
  L:=seq(plots[spacecurve]([[cos(k*Pi/n),sin(k*Pi/n),0],[cos(k*
  Pi/n),sin(k*Pi/n),f(cos(k*Pi/n),sin(k*Pi/n))]],
   color=navy_pâle,thickness=1),k=0..2*n):
  display([Surface,Contrainte,L,Contrainte_S,Domaine,
  Domaine_optimisation],axes=framed,
           style=patchnogrid,
           orientation=[62,69]);
```



En modifiant avec la souris l'orientation de cette surface, il est facile de se convaincre que la fonction f possède effectivement un maximun absolu et un minimum absolu sur ce domaine fermé. Mais, graphiquement, il est peut être même assez difficile de localiser sur la portion en cause de la surface, ces minimum et maximum absolus: sont-ils à l'intérieur ou sur le bord du domaine d'optimisation ?

I. Optima sur les points intérieurs du domaine: valeurs critiques

Commençons notre analyse avec les points intérieurs de ce disque. Recherchons donc les valeurs critiques de la fonction f.

Résolvons alors le système **S**: $\{f_x = 0, f_y = 0\}$

```
> fx:=D[1](f);
  fy:=D[2](f);
```

$$fx := (x, y) \mapsto 4 \cdot x$$

$$fy := (x, y) \mapsto 2 \cdot y - 1 \tag{3.2.2}$$

```
> solve({fx(x,y)=0,fy(x,y)=0},{x,y});
```

$$\left\{x = 0, y = \frac{1}{2}\right\} \tag{3.2.3}$$

Bien sûr, ce n'était pas un système très difficile à résoudre. Effectivement, on peut le faire sans l'aide de Maple, mais là n'est pas la question. On a donc à l'intérieur du disque, qu'une seule valeur critique. Nous contrôlerons ce point plus tard.

```
> Valeurs_critiques:=[0,1/2];
```

$$Valeurs\_critiques := \left[0, \frac{1}{2}\right] \tag{3.2.4}$$

II. Optima sur les valeurs frontières: méthode des multiplicateurs de Lagrange

Affichons seulement l'image de la frontière du disque.

```
> display([Contrainte,Domaine,Domaine_optimisation,Contrainte_S],
          axes=framed,
          style=patchnogrid,
          orientation=[35,65]);
```



En modifiant l'orientation du graphique avec la souris, il semble que cette fonction possède sur la frontière du disque, un maximum absolu atteint avec deux points (x, y) différents et un minimum absolu atteint avec un seul point.

Pour appliquer la méthode des multiplicateurs de Lagrange afin d'optimiser la fonction f sur le bord, nous limiterons donc le domaine d'optimisation de la fonction f à celui de la frontière du disque, soit le cercle de rayon 1. La contrainte $g(x, y) = 0$ dans la méthode des multiplicateurs de Lagrange sera donc l'équation du cercle unité.

```
> g:=(x,y)->x^2+y^2-1;
```

$$g := (x, y) \mapsto x^2 + y^2 - 1 \tag{3.2.5}$$

Formons maintenant la fonction F définie par $F(x, y) = f(x, y) + \lambda\, g(x, y)$. Le paramètre « λ » est appelé multiplicateur de Lagrange.

```
> F:=(x,y)->f(x,y)+lambda*g(x,y);
```

$$F := (x, y) \mapsto f(x, y) + \lambda \cdot g(x, y) \tag{3.2.6}$$

Résolvons maintenant le système S: $\{F_x(x, y) = 0, F_y(x, y) = 0, g(x, y) = 0\}$.

```
> Éq1:=diff(F(x,y),x)=0;
  Éq2:=diff(F(x,y),y)=0;
  Éq3:=g(x,y)=0;
  Sol:={solve({Éq1,Éq2,Éq3},{x,y,lambda})}; # Les accolades évitent
  la redondance des solutions
```

$$\acute{E}q1 := 2\lambda x + 4x = 0$$

$$\acute{E}q2 := 2\lambda y + 2y - 1 = 0$$

$$\acute{E}q3 := x^2 + y^2 - 1 = 0$$

$$Sol := \left\{\left\{\lambda = -2, x = -\frac{\sqrt{3}}{2}, y = -\frac{1}{2}\right\}, \left\{\lambda = -2, x = \frac{\sqrt{3}}{2}, y = -\frac{1}{2}\right\}, \left\{\lambda = -\frac{3}{2}, x = 0, y = -1\right\}, \left\{\lambda = -\frac{1}{2}, x = 0, y = 1\right\}\right\} \tag{3.2.7}$$

Énumérons toutes les valeurs de la frontière à contrôler. On pourrait, bien sûr, tout aussi bien établir cette liste manuellement.

```
> P:=Listage(Sol);
```

$$P := \left[-\frac{\sqrt{3}}{2}, -\frac{1}{2}\right], \left[\frac{\sqrt{3}}{2}, -\frac{1}{2}\right], [0, -1], [0, 1] \tag{3.2.8}$$

Ajoutons la valeur critique à la liste de ces valeurs à tester. Donnons à cette liste le nom Valeurs_C.

```
> Valeurs_C:=P,Valeurs_critiques;
```

$$Valeurs\_C := \left[-\frac{\sqrt{3}}{2}, -\frac{1}{2}\right], \left[\frac{\sqrt{3}}{2}, -\frac{1}{2}\right], [0, -1], [0, 1], \left[0, \frac{1}{2}\right] \tag{3.2.9}$$

Calculons les optima correspondant à ces cinq points.

```
> printf(`\n                         Valeurs à tester    |
  Valeurs de f    |`);
  printf(`\n                         ===================================
  ==============`);
  seq(printf(`\n  %40a   | %15a     |`, [Valeurs_C][k],f(op(1,op(k,
  [Valeurs_C])),op(2,op(k,[Valeurs_C])))),k=1..nops([Valeurs_C]));
```

```
                Valeurs à tester    |      Valeurs de f    |
==================================================
             [-1/2*3^(1/2), -1/2]   |          29/4      |
```

```
[1/2*3^(1/2), -1/2]   |          29/4      |
            [0, -1]   |             7      |
             [0, 1]   |             5      |
           [0, 1/2]   |          19/4      |
```

Reste finalement à conclure.

Le minimum absolu est atteint avec la valeur critique $\left(0, \frac{1}{2}\right)$ et vaut $\frac{19}{4}$ et le maximum absolu vaut $\frac{29}{4}$ et est atteint en $\left(\frac{\sqrt{3}}{2}, -\frac{1}{2}\right)$ et en $\left(-\frac{\sqrt{3}}{2}, -\frac{1}{2}\right)$.

Terminons cet exemple en illustrant ces optima sur la surface.

Remarquons que le minimum est atteint avec un point intérieur et que le maximum est atteint en deux points frontières.

```
> Points_optima:=[[0, 1/2,f(0,1/2)],[1/2*sqrt(3), -1/2,f(1/2*sqrt(3), -1/2)],[-1/2*sqrt(3), -1/2,f(-1/2*sqrt(3), -1/2)]];
```

$$Points\_optima := \left[\left[0, \frac{1}{2}, \frac{19}{4}\right], \left[\frac{\sqrt{3}}{2}, -\frac{1}{2}, \frac{29}{4}\right], \left[-\frac{\sqrt{3}}{2}, -\frac{1}{2}, \frac{29}{4}\right]\right] \quad \textbf{(3.2.10)}$$

```
> Points:=pointplot3d(Points_optima,symbol=solidsphere,symbolsize=15,color=navy):
display([Contrainte,Domaine,Domaine_optimisation,Surface,Points,Contrainte_S],lightmodel=none,style=patchnogrid,axes=framed,orientation=[70,55]);
```



Question: Peut-on imaginer ici, un disque fermé de rayon r centré au point $(0, 0, 0)$ dans le plan Oxy dont le rayon sera assez petit de telle sorte que la fonction ne puisse posséder de points critiques à l'intérieur de ce disque ? Si oui, cela constiturerait un exemple de fonction où les extréma absolus seraient situés exclusivement sur la frontière de ce domaine fermé.

Exemple 2

Reprenons avec Maple un problème que nous avons déjà traité en classe.

Trouver les distances maximum et minimum entre l'origine et la courbe d'équation $5x^2 + 6xy + 5y^2 = 8$.

Soit D la distance entre un point (x, y) de la courbe d'équation $5x^2 + 6xy + 5y^2 = 8$ et l'origine $(0, 0)$.
Alors $D = \sqrt{(x-0)^2 + (y-0)^2} = \sqrt{x^2 + y^2}$.

Optimiser le carré de la distance D^2 est aussi optimiser la distance D puisque $D > 0$.

Soit donc la fonction f définie par $f(x, y) = x^2 + y^2$ à optimiser.

```
> f:=(x,y)->x^2+y^2;
```

$$f := (x, y) \mapsto y^2 + x^2 \tag{3.3.1}$$

On recherche les extréma absolus de la fonction f sous la contrainte $5x^2 + 6xy + 5y^2 = 8$.

I. Optima sur les points intérieurs

Le domaine d'optimisation étant une égalité, soit le lieu des valeurs (x, y) vérifiant l'équation $5x^2 + 6xy + 5y^2 = 8$, il n'y a aucun point intérieur à cette région. Il n'y a donc aucun point critique à contrôler.

II. Optima sur les valeurs frontières: méthode des multiplicateurs de Lagrange

Appliquons la procédure des multiplicateurs de Lagrange à la fonction f définie par $f(x, y) = x^2 + y^2$.
Formons la fonction $F(x, y) = f(x, y) + \lambda\, g(x, y)$.

```
> g:=(x,y)->5*x^2+6*x*y+5*y^2-8;
  F:=(x,y)->f(x,y)+lambda*g(x,y);
```

$$g := (x, y) \mapsto 5 \cdot x^2 + 6 \cdot y \cdot x + 5 \cdot y^2 - 8$$

$$F := (x, y) \mapsto f(x, y) + \lambda \cdot g(x, y) \tag{3.3.2}$$

La règle de F est donc:

```
> F(x,y);
```

$$x^2 + y^2 + \lambda\,(5x^2 + 6xy + 5y^2 - 8) \tag{3.3.3}$$

Résolvons maintenant le système S: $\{F_x(x, y) = 0, F_y(x, y) = 0, g(x, y) = 0\}$.

```
> Éq1:=diff(F(x,y),x)=0;
  Éq2:=diff(F(x,y),y)=0;
  Éq3:=g(x,y)=0;
  Sol:={solve({Éq1,Éq2,Éq3},{x,y,lambda})};
```

$$\acute{E}q1 := 2x + \lambda\,(10x + 6y) = 0$$

$$\acute{E}q2 := 2y + \lambda\,(6x + 10y) = 0$$

$$\acute{E}q3 := 5x^2 + 6xy + 5y^2 - 8 = 0$$

$$Sol := \left\{ \left\{ \lambda = -\frac{1}{2}, x = \sqrt{2}, y = -\sqrt{2} \right\}, \left\{ \lambda = -\frac{1}{2}, x = -\sqrt{2}, y = \sqrt{2} \right\}, \left\{ \lambda = -\frac{1}{8}, x = -\frac{\sqrt{2}}{2}, y = -\frac{\sqrt{2}}{2} \right\}, \left\{ \lambda = -\frac{1}{8}, x = \frac{\sqrt{2}}{2}, y = \frac{\sqrt{2}}{2} \right\} \right\} \tag{3.3.4}$$

```
> Valeurs_C:=Listage(Sol);
```

(3.3.5)

$$Valeurs\_C := \left[\sqrt{2}, -\sqrt{2}\right], \left[-\sqrt{2}, \sqrt{2}\right], \left[-\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2}\right], \left[\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}\right] \quad \textbf{(3.3.5)}$$

Calculs des valeurs de la fonction f en ces points.

```
> printf(`\n                          Valeurs à tester     |   Valeurs
  de f  |`);
  printf(`\n            ============================================
  ========`);
  seq(printf(`\n %40a    | %10a      |`, [Valeurs_C][k],f(op(1,op(k,
  [Valeurs_C])),op(2,op(k,[Valeurs_C])))),k=1..nops([Valeurs_C]));
```

```
                  Valeurs à tester     |   Valeurs de f  |
==================================================
            [2^(1/2), -2^(1/2)]    |         4      |
            [-2^(1/2), 2^(1/2)]    |         4      |
  [-1/2*2^(1/2), -1/2*2^(1/2)]    |         1      |
    [1/2*2^(1/2), 1/2*2^(1/2)]    |         1      |
```

Reste alors à conclure.

Avec la contrainte $5x^2 + 6xy + 5y^2 = 8$, la valeur maximale de $f(x, y)$ est 4 et la valeur minimale de $f(x, y)$ est 1:

Ainsi, la valeur maximale de D^2 est 4 et la valeur minimal de D^2 est 1.
la valeur maximale de D est 2 et la valeur minimale de D est 1.

Donc, les distances maximum et minimum entre l'origine et la courbe d'équation $5x^2 + 6xy + 5y^2 = 8$, sont respectivement 2 unités et 1 unité.

La macro-commande *extrema*

La macro-commande extrema de la bibliothèque principale applique la méthode des multiplicateurs de Lagrange pour optimiser une fonction avec une contrainte d'égalité.

Illustrons l'utilisation de cette macro-commande qui automatise la recherche des optima de $f(x, y) = 2x^2 - y + y^2 + 5$ avec la contrainte $g(x, y) = x^2 + y^2 - 1$.

```
> f:=(x,y)->2*x^2-y+y^2+5;
  g:=(x,y)->x^2+y^2-1;
```

$$f := (x, y) \mapsto 2 \cdot x^2 - y + y^2 + 5$$

$$g := (x, y) \mapsto x^2 + y^2 - 1 \quad \textbf{(4.1)}$$

```
> extrema(f(x,y),g(x,y)=0);
```

$$\left\{5, \frac{29}{4}\right\} \quad \textbf{(4.2)}$$

Pour obtenir le détail des couples (x, y) permettant d'atteindre ces optima (avec la contrainte en cause), il faut donner à la macro-commande deux arguments supplémentaires. D'abord l'ensemble des variables impliquées dans les équations et un nom Maple pointant vers l'ensemble-solution du système d'équation que cette macro-commande résoud. Il est recommandé de mettre entre apostrophes droits ce nom car au moment de l'appel, nom doit être une variable libre.

```
> extrema(f(x,y), g(x,y)=0, {x,y}, 'Sol');
```

$$\left\{5, \frac{29}{4}\right\} \quad \textbf{(4.3)}$$

```
> Sol;
```

$$\left\{\{x=0, y=-1\}, \{x=0, y=1\}, \left\{x=-\frac{\sqrt{3}}{2}, y=-\frac{1}{2}\right\}, \left\{x=\frac{\sqrt{3}}{2}, y=-\frac{1}{2}\right\}\right\} \quad \textbf{(4.4)}$$

ATTENTION: Pour obtenir une approximation décimale de chaque élément de solution dans le cas où les racines sont formulées avec la macro-commande RootOf, il faudra énoncer la requête evalf(allvalues(Sol)) et non pas evalf(Sol).

```
> evalf(Sol);
```

$$\{\{x=-0.866025404, y=-0.5\}, \{x=0., y=-1.\}, \{x=0., y=1.\}, \{x=0.866025404, y=-0.5\}\} \quad \textbf{(4.5)}$$